FUNCIÓN REAL DE VARIABLE REAL
APLICACIONES PRÁCTICAS

EJERCICIO 1

Se desea construir una rampa mecánica para unir tres plantas de un edificio según la figura



La altura de cada planta es de 5 metros y cada uno de los 3 tramos horizontales es de 3 metros. La rampa se eleva verticalmente 1 metro por cada 3 metros horizontales.

a) Definir una función que nos de la altura "y" en función de "x".

b) Hallar la longitud total de la rampa. Si esta avanza 0.3 m/seg, ¿cuánto se tarda en llegar desde la planta baja a la segunda planta?.

▼ Apartado a

⇒ Definición de la función

```
f1[x_] = (x - 3) / 3
```

$$\frac{1}{3}(-3+x)$$

```
f2[x_] = 1 / 3 * (x - 21) + 5
```

$$5+\frac{1}{3}(-21+x)$$

```
f[x_] = Which[0 ≤ x ≤ 3, 0, 3 < x < 18, f1[x],
  18 ≤ x ≤ 21, 5, 21 < x < 36, f2[x], 36 ≤ x ≤ 39, 10]
```

```
Which[0 ≤ x ≤ 3, 0, 3 < x < 18, f1[x], 18 ≤ x ≤ 21, 5, 21 < x < 36, f2[x], 36 ≤ x ≤ 39, 10]
```

⇒ **Gráfica de la función**

```
Plot[f[x], {x, 0, 39},
 Ticks -> {{0, 3, 18, 21, 36, 39}, {0, 5, 10}}, AxesLabel → {x, y}]
```



Apartado b

⇒ **Longitud total de la rampa**

```
d = 3 * 3 + 2 * Sqrt[15^2 + 5^2]
```

$9 + 10\sqrt{10}$

```
d // N
```

```
40.6228
```

⇒ **Tiempo**

En segundos

```
t = d / 0.3
```

```
135.409
```

En minutos

```
t / 60
```

```
2.25682
```

aproximadamente 2 minutos y 15 segundos

EJERCICIO 2

La función parte entera de un número real x se define como E|x| = número mayor *n* tal que *n* ≤ x. Utilizar dicha función para describir el coste C de una llamada en función del número *t* de minutos empleados, sabiendo que una llamada telefónica interurbana cuesta 1.4 euros los dos primeros minutos y 0.8 euros cada minuto adicional o fracción. Dibujar la gráfica de la función coste.

⇒ Definición de la función parte entera

La instrucción Floor devúelve el mayor entero menor o igual que x

```
g[x_] = Floor[x];
```

```
Plot[Floor[x], {x, -3, 3}, PlotStyle → Thickness[0.01]]
```



⇒ Definición de la función coste

Si 0 ≤ t < 2, definimos la función C = 1,4 y si t ≥ 2, utilizamos la función parte entera para añadir al coste los 0,8 euros de cada minuto adicional o fracción

```
coste1[x_] = 1.4;
coste2[x_] = coste1[x] + 0.8 Floor[x - 1];
```

⇒ **Representación gráfica de la función coste**

```
g1 = Plot[coste1[x], {x, 0, 2}, PlotStyle → Thickness[0.01]];
g2 = Plot[coste2[x], {x, 2, 8}, PlotStyle → Thickness[0.01]];
Show[g2, g1, PlotRange → {0, 7}, AxesOrigin → {0, 0},
 AspectRatio → Automatic, AxesLabel → {tiempo, coste}]
```



EJERCICIO 3

a) Definir las funciones sen(x), sen(x) + 1 y sen(x) + 2.
b) Representar las funciones sobre los mismos ejes haciendo variar el rango de la abscisa entre - π y π y rellenando de un color el espacio entre la primera y la segunda función y de otro color el espacio entre la segunda y la tercera función.
c) Representar las funciones definidas en el apartado a) una a continuación de otra en una fila.
d) Representar las funciones definidas en el apartado a) una a continuación de otra en una columna.

▼ Apartado a

⇒ **Definición de las funciones**

```
f1[x_] = Sin[x]; f2[x_] = Sin[x] + 1; f3[x_] = Sin[x] + 2;
```

Apartado b

⇒ Representación gráfica de las tres funciones sobre los mismos ejes

```
Plot[{f1[x], f2[x], f3[x]}, {x, -Pi, Pi},
 PlotStyle → Thickness[0.02], Filling → {{1 -> {2}}, {2 → {3}}}]
```



Apartado c

⇒ Representación gráfica de las tres funciones en una fila

```
grafico1 = Plot[f1[x], {x, -Pi, Pi},
   PlotStyle → {Thickness[0.015], Purple}];
grafico2 = Plot[f2[x], {x, -Pi, Pi},
  PlotStyle → {Thickness[0.015], Orange}]; grafico3 =
 Plot[f3[x], {x, -Pi, Pi}, PlotStyle → {Thickness[0.015], Blue}];
```

```
GraphicsGrid[{{grafico1, grafico2, grafico3}}]
```



Apartado d

⇒ Representación gráfica de las tres funciones en una columna

```
grafico1 = Plot[f1[x], {x, -Pi, Pi},
   PlotStyle → {Thickness[0.015], Purple}];
grafico2 = Plot[f2[x], {x, -Pi, Pi},
  PlotStyle → {Thickness[0.015], Orange}]; grafico3 =
 Plot[f3[x], {x, -Pi, Pi}, PlotStyle → {Thickness[0.015], Blue}];
```

```
GraphicsGrid[{{grafico1}, {grafico2}, {grafico3}}, Frame → True]
```



EJERCICIO 4

a) Definir las funciones f(x,y) = sen(x)sen(y) - 0,5 y g(x,y) = cos(x)cos(y) - 0,5.
b) Representar las funciones sobre los mismos ejes, asignándoles colores diferentes y coloreando asimismo el fondo del gráfico.

Apartado a

⇒ **Definición de las funciones**

```
f[x_, y_] = Sin[x] * Sin[y] - 0.5;
g[x_, y_] = Cos[x] * Cos[y] - 0.5;
```

Apartado b

⇒ **Representación gráfica de las funciones**

```
ContourPlot[{f[x, y] == 0, g[x, y] == 0}, {x, -2 π, 2 π},
 {y, -2 π, 2 π}, ContourStyle → {{Thickness[0.01], Yellow},
   {Thickness[0.01], Green}}, Background → LightBlue]
```



EJERCICIO 5

Escribe las instrucciones necesarias para obtener el siguiente gráfico:



⇒ **Definición del cuadrado**

```
puntos = {{-3, -3}, {3, -3}, {3, 3}, {-3, 3}, {-3, -3}};
cuadrado = ListPlot[puntos, Joined → True, AspectRatio → Automatic]
```



⇒ **Definición del círculo**

```
circulo =
 ContourPlot[x^2 + y^2 == 4, {x, -2, 2}, {y, -2, 2}, Frame → None]
```



⇒ Definición de la recta

```
puntos = {{-2, 0}, {2, 0}};
linea = ListPlot[puntos, Joined → True]
```



⇒ Combinación de los tres gráficos

```
Show[cuadrado, circulo, linea, Axes → False]
```